/*
Author: Andrew Reed
Proposal title: When is "bad stronger than good" in decision making

HYPOTHESES


**Stated-Hyp1: "we predict that in the low-stakes condition older (vs. younger)
adults will give more weight to positive (vs. negative) reviews" (p. 2)

	Test-Hyp1: Within low-stakes condition, the effect of positive vs. negative reviews on percieved informativeness will be moderated by age.
	
	Test-Hyp2: Within low-stakes condition, the effect of positive vs. negative reviews on percieved trustworthiness will be moderated by age.
	
	
**Stated-Hyp2: "we predict that in the high-stakes condition, older and younger adults will weight reviews similarly." (p. 2)

	Test-Hyp3: Within high-stakes condition, the effect of positive vs. negative reviews on percieved informativeness will NOT be moderated by age.
	
	Test-Hyp4: Within high-stakes condition, the effect of positive vs. negative reviews on percieved trustworthiness will NOT be moderated by age.
	
	
********************************************************************************
NOTES:
- For stated hypothesis 1, the outcome seems to be attentiveness to information. 
However, there is no measure of attentiveness in the study. There is however a 
measure of percieved infromativeness. We are assuming that the author meant the 
latter. 	
*/

clear all
use "ReedB32.dta", clear

********************************************************************************

* TREATMENT

* Stake condition (high vs. low stakes medical procedure)
	tab xTess166
	recode xTess166 (1 3=1 "high stakes") (2 4=0 "low stakes"), gen(highstakes)
	tab highstakes, mis

* valence (positive vs. negative physican review)
	recode xTess166 (1 2=1 "positive") (3 4=0 "negative"), gen(positivevalence)
	tab positivevalence, mis

* Age
	tab PPAGE, mis // no missing data
	clonevar age=PPAGE

* OUTCOMES

* informativeness [How helpful is this review?]
	tab QR1
	clonevar informative=QR1 
	replace informative= .r if informative==-1
	tab informative, mis 
// higher values indicate more informativeness

* trustworthiness  [How trustworthy is the person who wrote this review?]
	tab QR2
	clonevar trustworthy=QR2
	replace trustworthy=.r if trustworthy==-1
	tab trustworthy, mis 
// higher values indicate greater trustworthiness


********************************************************************************

* ANALYSIS


*Test-Hyp1: Within low-stakes condition, the effect of positive vs. negative reviews on percieved informativeness will be moderated by age.
	reg informative i.positivevalence##c.age if highstakes==0
	// reject. p=0.195
	tess 1.positivevalence#c.age, bonf(2) init(ReedB32)
	
*Test-Hyp2: Within low-stakes condition, the effect of positive vs. negative reviews on percieved trustworthiness will be moderated by age.
	reg trustworthy i.positivevalence##c.age if highstakes==0
	// reject. p=0.170
	tess 1.positivevalence#c.age, bonf(2)
	
*Test-Hyp3: Within high-stakes condition, the effect of positive vs. negative reviews on percieved informativeness will NOT be moderated by age.
	reg informative i.positivevalence##c.age if highstakes==1
	// do not reject. p=0.090
	tess 1.positivevalence#c.age, bonf(2)
	
*Test-Hyp4: Within high-stakes condition, the effect of positive vs. negative reviews on percieved trustworthiness will NOT be moderated by age.
	reg trustworthy i.positivevalence##c.age if highstakes==1
	// do not reject. p=0.987
	tess 1.positivevalence#c.age, bonf(2)			
